
Abstract
The APL notation would appear to be a clear match for con-
volutional neural networks, but traditional implementations
of APL have lagged behind the performance of highly tuned,
specialized frameworks designed to execute CNNs on the
GPU. Moreover, most demonstrations of APL for neural net-
working have involved relatively small examples. We ex-
plore a more complex example in the U-net architecture and
utilize a modern APL compiler with GPU support, Co-dfns,
to compare the state of the art of APL against the current
crop of specialized neural network frameworks in the form
of PyTorch. We compare performance as well as the lan-
guage design of APL for neural network programming and
the clarity and transparency of the resulting code.

We found that the complete “from scratch” APL source
was on par with the complexity of the PyTorch reference
implementation, albeit more foreign, while being more con-
cise and complete. We also found that when compiled with
Co-dfns, despite the naive implementation both of Co-dfns
and our own code, performance on the GPU and the CPU
were within a factor of 2.2 - 2.4 times that of the PyTorch im-
plementation. We believe this suggests significant avenues
of future exploration for machine learning language design,
pedagogy, and implementation, both inside and outside of
the APL community.

1 Introduction
Specializedmachine learning frameworks dominate the present
industrial and educational spaces for deep learning applica-
tions. A wide number of higly specialized and highly op-
timized libraries exist, often built on top of one another,
to support the modern wave of machine learning architec-
tures. These systems are often more complex than your
typical library, and they might even be better classified as
their own domain-specific languages (DSLs). While these
libraries have supported the current explosion of machine
learning developers, a number of issues have emerged.

First, because of their highly specialized nature, users of
these systems tend to become experts not in generalized
programming or algorithmic skills, but specialist toolkits
and frameworks around a very specific model of compu-
tation. This specialized nature often mandates dedicated
courses and even entire academic specializations (even at
the undergraduate level) focused on the mastery of these
particular concepts. This can create a sharp fall off of skills
transferrence, where machine learning experts can use ma-
chine learning frameworks effectively, but may be underde-
veloped and underprepared to handle situations that require
a broader or more adaptive skillset.1

Second, from a pedagogical perspective, when teaching
machine learning, one may often be able to implement sim-
ple networks in a general-purpose programming language,
but trying to teach machine learning through a typical gen-
eral purpose language can be difficult, because one quickly
encounters performance and scalability limitations thatmake
any non-trivial and interesting applications likely beyond
the competency and endurance of your typical student. This
creates a sharp contrast in which one begins with simple
systems that can be programmed “by hand” but quickly tran-
sitions to highly opaque and complex frameworks that are
difficult to understand, modify, or intuit. This can result in
significant reductions in professional competency.

Third, if a lack of profound and intuitive understanding
of the underlying mechanics of a deep learning system con-
tinues into professional life, the result can be a type of “pro-
gramming by knob turning” in which neural networks are
programmed via trial and error rather than through inten-
tional design. Machine Learning as a discipline is already

1To someone who only has a hammer, everything looks like a nail.
1

opaque enough, withmany cases of unintended consequences,
without the added dangers inherent in this sort of uninten-
tional programming guesswork [Domingos 2012].

Fourth, the specificity of machine learning frameworks
can result in significant amounts of code churn and a reduc-
tion in the stability of codebases for enterprise use. Switch-
ing hardware, architectures, operating systems, or the like
can create unstable conditions inwhich codemust be rewrit-
ten, adapted, or thrown away entirely. Machine learning
frameworks are often highly vendor-specific, and even those
which are more vendor-neutral tend to encode significantly
greater specificity than is historically warranted for code in-
tended to last for any long period of time. This almost ne-
cessitates higher levels of programmer investment in order
to keep such systems running over a long period of time.

Despite the above potential issues, specialist frameworks
have proven highly effective, in large part because of how
important high-performance is to the domain of machine
learning. However, in recent years, general-purpose array
programming languages have seen a resurgence, and nat-
urally, they have been examined in the light of machine
learning. Such languages were also popular during early ex-
ploration of neural network programming during the 20th
century [Alfonseca 1990], but performance issues of then-
current hardware prevented further progression.

APL, as a general-purpose array programming language,
created by Kenneth Iverson as an improved mathematical
notation [Iverson 1962], has seen an increase in popularity
over the past decades, in part because of the renewed inter-
est in parallel computation and a wider acceptance of the
use of a variety of programming languages. However, only
recently has significant new research into the use of APL as
a possible implementation language for machine learning
begun to surface.

The long history of APL, its origins as an pedagogical tool,
and its reputation for directness of algorithmic expression
[Knuth 1993, 2007] help to address some of the concerns
above. Furthermore, it is one of the most linguistically sta-
ble languages, while also being exceptionally high level and
high performance at the same time [Hsu 2019], making it
highly suitable for long lived code as well as rapid prototyp-
ing. Finally, the language itself defaults to a data-parallel
semantics, making its application to GPU programming an
obvious conclusion.

While the above advantages might suggest APL as a ter-
rific tool for machine learning, unfortunately, the vast ma-
jority of implementations have been for the CPU only, and
those have usually been entirely interpreted. Traditionally,
compiler implementors have considered APL a challenging
language to compile [Hsu 2019], but recent innovations to
the language (particularly those with a functional program-
ming focus) have made compilation much more tractable,
and the Co-dfns compiler now exists as an APL implemen-
tation with native GPU support [Hsu 2019].

Given the available APL technology and the parsity of ex-
isting materials on modern machine learning development
in APL, we conducted an exploration into the state of the art
in APL, both from a language design and a runtime imple-
mentation perspective. To do this, we focused our efforts
on the implementation and benchmarking of the U-net con-
volutional neural network [Ronneberger et al. 2015]. This
is a popular image segmentation architecture with a partic-
ularly interesting U-shaped design. It makes use of a range
of popular CNN vocabularies and functions while having a
clear architecture that is not so simple as to be trivial. This
makes it an ideal candidate for exploring APL’s capabilities.

We make the following contributions:

• A complete demonstration in APL of the popular U-
net convolutional neural network, which is non-trivial
in vocabulary and architecture
• OurU-net implementation is exceptionally simple, con-

cise, transparent, and direct
• Our implementation was written with pure APL and

no dependencies, frameworks, libraries, or other sup-
porting code outside of the APL implementation
• A functional programming-friendly approach to neu-

ral network design and implementation
• An analysis and examination of the current language

features within APL that appear relevant to CNNs
and machine learning
• A critical discussion and design comparison of two

different approaches to supporting convolutions and
similar operations in a general-purpose array language
with a recommendation for future implementation im-
provements
• Agrounded perspective on the applications of general-

purpose array programming languages like APL to
themachine learning space fromprofessional and ped-
agogical angles and howAPL compares to alternative,
specialist framework approaches
• Performance results of two modern APL implemen-

tations, one compiled and the other interpreted, on
CPU and GPU hardware against a reference PyTorch
implementation for U-net
• Performance observations of specialized neural net-

work functionality exposed in more general purpose
array frameworks for GPU programming
• Specific highlighting of low-hanging fruit for improv-

ing the current range of APL implementations both
in terms of language design and runtime implemen-
tation
• Ademonstration of the expressiveness and performance

that careful language design can enable without the
need for complex implementation models or theory

2

2 Background
In this section, we provide the relevant background pertain-
ing to the machine learning concepts needed to work with
CNNs, and the u-net in particular, and to the APL language.

2.1 Convolutional Neural Networks
The experiment this paper uses to produce its benchmarks
is the reproduction of a well-known convolutional neural
network architecture. The use of CNNs in machine learn-
ing was widely popularised with the publication of a paper
[Krizhevsky et al. 2012] that used CNNs to achieve state-
of-the-art performance in labeling pictures of the ImageNet
[Deng et al. 2009] challenge. However, a proeminent paper
from 1998 [LeCun et al. 1998] shows that the modern use of
CNNs can be dated farther back.

The use of convolutional neural networks, as we know
them today, builds on top of the convolutional layer [O’Shea
andNash 2015]. Convolutional layers receive three-dimensional
tensors as input and produce three-dimensional tensors as
output. These inputs have a fixed number of channels2 𝑛𝑖𝑛
which are then transformed into𝑛𝑜𝑢𝑡 channels throughmeans
of discrete convolutions with a total of 𝑛𝑖𝑛 × 𝑛𝑜𝑢𝑡 kernels,
the learnable parameters of the convolutional layer. One of
the advantages of CNNs is that, although the total number
of kernels 𝑛𝑖𝑛 × 𝑛𝑜𝑢𝑡 depends on the number of input and
output channels, the sizes of the kernels are independent of
the size of the other two dimensions of the inputs. Despite
the fact that the main dynamics of a convolutional layer
is governed by discrete convolution with the learnable ker-
nels, the exact behaviour of a convolutional layer depends
on layer parameters like the padding and the stride used
[Dumoulin and Visin 2016].

Given that CNNswere primarily used in image recognition-
related tasks, convolutional layers were often paired with
pooling layers that ease the recognition of features over small
neighbourhoods [Scherer et al. 2010]. The rationale behind
these pooling layers, as seen from an image recognition-
related context, can be interpreted as follows: the image
features one is typically interested in (e.g., the recognition
or segmentation of objects, or image labeling) are not con-
tained in single pixels of the input images, but in regions
of said pixels. Pooling layers are, thus, employed with the
purpose of aggregating low-level information that can then
be used to recognise the larger features of interest [Scherer
et al. 2010].

In 2015, three authors published a paper [Ronneberger
et al. 2015] introducing the u-net architecture: a CNNwith a
non-trivial architecture that won several biomedical image
segmentation challenges at the time of its publication. Since
then, the u-net architecture was reimplemented hundreds

2“channel” typically refers to the leading dimension of these inputs/outputs,
a nomenclature that is derived from the fact that CNNs were popularised
in the context of image processing.

of times3, most notably through the use of deep-learning
frameworks such as PyTorch [Paszke et al. 2019], a deep-
learning framework used in this work, or Caffe [Jia et al.
2014], which is the deep learning framework in which the
original u-net was implemented. For this paper, we reimple-
mented the u-net architecture, in APL, without making use
of any (machine learning) libraries or frameworks. Before
we introduce our work on that implementation, we discuss
the original architecture that we set out to replicate.

2.2 Original U-net Architecture
Figure 1 shows the original diagram that represents the u-
net architecture [Ronneberger et al. 2015], which we cover
now. We will go through the figure from left to right, fol-
lowing the U-shape of the diagram.

The blue right arrows, labeled “conv 3x3, ReLU”, repre-
sent unpadded convolutions with 3 × 3 kernels. Figure 1
shows that after each of these convolutions, the size of the
feature maps decreases by 2, from which it can be inferred
that the stride [Dumoulin and Visin 2016] is 1. After each
convolution, we use the activation function rectified linear
unit (ReLU) [Nwankpa et al. 2018]. Pairs of these convolu-
tions are followed by max-pooling operations represented
by the red down arrows. These max-pooling operations act
on a 2×2 region and have a stride of 2, effectively downsam-
pling each feature map to half the size. Because of this re-
peated halving, the input size must be chosen carefully4. Af-
ter every downsampling step, the first convolution doubles
the number of channels. The pattern of two convolutions
(with ReLUs) followed by downsampling via max-pooling
happens four times and makes up the contracting path of
the network, on the left of Figure 1.

Having reached the end of the contracting path (at the
bottom of the diagram), we start the expanding path. The
expanding path also makes use of unpadded convolutions
with 3× 3 kernels and stride 1, but these are now at the end
of each step, instead of at the beginning. Each step of the ex-
panding path starts with an upsampling operation (green up
arrows) that doubles the size of the feature maps while cut-
ting their number down in half. For this upsampling, we in-
fer that the original authors used a transposed convolution
(with 2 × 2 kernels) of stride 2 [Dumoulin and Visin 2016]5.
These transpose convolutions produce half of the channels
that are fed as input to the regular convolutions. The other
half of the channels is copied and cropped from the corre-
sponding step in the contracting path, as represented by the
gray right long arrows in the middle of the diagram of Fig-
ure 1. Because there is a slight mismatch between the size of
the feature maps that are copied and cropped and the other
feature maps that resulted from the upsampling step, the

3Numbers by Papers with Code as of March, 2022.
4Specifically, the input dimensions must be congruent to 12 mod 16
5See [Serrão 2022] for an informal discussion of this inferrence.

3

Figure 1. Original u-net architecture, as seen in the original paper [Ronneberger et al. 2015]. Arrows represent operations
between the multi-channel feature maps represented by the rectangles. The number on top of each rectangle is its number of
channels and the numbers in the lower-left corner are the 𝑥 and 𝑦 dimensions of the feature maps.

feature maps that are copied get cropped from the centre of
the larger feature maps of the contracting path. It is after
this copy and crop operation that we feed the feature maps
into the two convolution layers that are paired with their
respective ReLU activation functions.

At the end of the contracting path, we have a 1 × 1 un-
padded convolution that reduces the 64 feature maps to 2
feature maps (one per class).

To compute the loss of the output with respect to the ex-
pected labels, we compute the softmax across the two out-
put channels followed by the cross entropy loss function.

2.3 APL Notation
APL [Iverson 1962] is an alternative mathematical notation,
introduced by Turing award winner Kenneth E. Iverson in
the ’60s, that has since evolved into an executable mathe-
matical notation [Hui and Kromberg 2020]. In this section,
we introduce the basics of the APL notation, but the reader
is directed to [Legrand 2009] for a full tutorial. Online inter-
active systems are also available6, which should make it eas-
ier to get acquainted with APL. Throughout the remainder
of this paper, the notation used is such that it is compatible
with Dyalog APL 18.07.

6TryAPL https://tryapl.org is an example of such a service.
7You can get Dyalog APL from Dyalog’s website https://dyalog.com/
download-zone.htm.

2.3.1 Functions andArrays. APL is an “alternative”math-
ematical notation because it differs from the traditionalmath-
ematical notation in some ways. However, not everything
in APL is foreign, as demonstrated by the following exam-
ples of addition and multiplication:

1 + 2
3

73 × 104
7592
The format of the two examples abovewill be the same through-
out the paper8: the notation typed by the user is indentend
to the right and the computed result is left-aligned on the fol-
lowing line(s). Subtraction and division are also represented
by the usual glyphs, - and ÷, respectively:

10 - 1 2 3
9 8 7

100 50 20 ÷ 2
50 25 10
In APL, one is allowed to write multiple values next to each
other, which are then stranded together and interpreted as a
vector. Thus, 1 2 3 represents the three-item vectorwhose
elements are the first three positive integers. Then, the APL
function minus takes the scalar 10 as its left argument and
the three-item vector 1 2 3 as its right argument, and it
subtracts each of the items of the right argument vector
8This format mimics that of the APL session, the interactive environment
in which one can use APL.

from its left argument. Similarly, the division example shows
that vectors can also be used as the left argument. The natu-
ral progression is to wonder whether vectors can be used on
the left and on the right of a function, and typically they can.
We demonstrate that with themax function, represented by
the upstile glyph ⌈:

(1⌈5) (10⌈5) (100⌈500) (1000⌈500)
5 10 500 1000

1 10 100 1000 ⌈ 5 5 500 500
5 10 500 1000

The first example shows how parenthesis () can be used
to create vectors whose items are results of other expres-
sions, given that the four expressions inside parenthesis pro-
duced the four items of the result vector. The second exam-
ple shows that we can obtain the same result by collecting
all the left arguments inside all () on the left of a single ⌈,
and by collecting all the right arguments inside all () on
the right of that same ⌈.

The dyadic functions plus +, minus -, times ×, divide ÷,
and max ⌈, all share the property that allows them to ac-
cept vectors as arguments: they are scalar functions. Scalar
functions are functions that pervade the structure of the
argument(s) and apply directly to each of the scalars that
make up said argument(s). This becomes increasingly rel-
evant when one understands that APL has first-class sup-
port for arrays of any dimension, of which we have seen
scalars such as 10 and 73 and vectors. Scalars and vectors
can be typed directly but arrays of higher dimensions must
be loaded from an external data source or created dynami-
cally through computations.

The reshape function is represented by the Greek letter
rho ⍴ and is a dyadic function that reshapes its right argu-
ment to have the shape specified by the left argument. For
instance, if we want to create a 2 × 3 matrix with the first
six non-negative integers, we can do it like so:

2 3 ⍴ 0 1 2 3 4 5
0 1 2
3 4 5

Each non-negative integer of the left argument specifies the
length of the result along the corresponding dimension. So,
if the left argument had been 5 9 7, the resulting array
would have been a cuboid (array with three dimensions)
composed of 5 planes, 9 rows, and 7 columns, holding a
total of 5 × 9 × 7 = 315 items.

Given an arbitrary arrayarray, we can also use theGreek
letter rho ⍴ to compute the shape of the array, that is, the
length of each of its axis, or dimensions. In the example be-
low, we can see that array is a matrix with 2 rows and
3 columns, even though we don’t know what the items of
array are:

⍴array
2 3

This also goes to show that many functions have two be-
haviours, one monadic behaviour and one dyadic behaviour.
A function is used monadically when it has an array argu-
ment on its right, but not on its left, and a function is used
dyadically when it has an array argument on its left and an-
other one on its right. For example, rho ⍴ represents the
monadic function shape and the dyadic function reshape. In
this particular instance, we can also see that array is a
nested matrix:

array
┌───┬───┬───┐
│0 0│0 1│0 2│
├───┼───┼───┤
│1 0│1 1│1 2│
└───┴───┴───┘

Thecells above each contain a two-item vector (0 0 through
1 2), and the borders surrounding those two-item vectors
are a visual cue to help the reader discern the nested nature
of the array.

Another key difference between the APL notation and the
traditional mathematical notation is that APL normalises
precedence rules by saying that all functions have the same
precedence: functions are said to have a long right scope and
a short left scope, which is why APL is often said to “execute
from right to left”. A long right scope means that a function
takes as right argument everything to its right, whereas a
short left scope means that a function only takes as left argu-
ment the array that is immediately to its left. The expression
2×3−4×5, in standard mathematical notation, is equivalent
to (2 × 3) − (4 × 5) = 6 − 20 = −14, because multiplication
has higher precedence over subtraction. However, the APL
expression 2×3-4×5 is equivalent to 2×(3-(4×5)):

(2 × 3) - (4 × 5)
¯14

2 × (3 - (4 × 5))
¯34

2 × 3 - 4 × 5
¯34

APL uses the high-minus ¯ to represent negative numbers,
otherwise there would be ambiguity in the use of the minus
sign -9.

2.3.2 Shape, Rank, Data. Every APL array can be funda-
mentally characterised by its shape, its rank, and its data:
• the shape of an array can be computed with the func-

tion shape and is a vector that specifies the length of
each dimension of its argument;
• the rank of an array is the number of its dimensions

(the length of its shape) and dictates the name of said
array as per Table 1; and

9Is 1 -2 the APL expression “one minus two” or the two-item vector “one,
negative two”?

Rank Name

0 scalar
1 vector
2 matrix
3 cuboid

Table 1. Array names according to rank.

• the data of an array are the items that compose said
array.

The shape of an array arr is ⍴arr. The rank of an array
is the length of its shape or, in APL vocabulary, the tally
of its shape, which is ≢⍴. Finally, the data of an array can
be retrieved as a vector with the monadic function ravel ,
(comma). These are illustrated below for a matrix. We use
the primitive roll ? to fill the matrix with random data and
annotate the expressions with APL comments ⍝:

⍝ Create random array mat
mat ← ?2 3⍴0
mat

0.999 0.00424 0.351
0.967 0.92 0.821

⍝ mat has shape 2 3.
⍴mat

2 3
⍝ mat has rank (tally shape) 2
≢⍴mat

2
⍝ mat contains this data:
,mat

0.999 0.00424 0.351 0.967 0.92 0.821

2.3.3 Operators. On top of providing a rich set of built-in
functions, APL provides a series of operators that allow us
to combine and modify our functions. A typical example of
a monadic APL operator is reduce-first ⌿. Themonadic oper-
ator reduce-first takes a function on its left and then inserts
it between the elements of the right argument. Previously,
we computed the total number of elements in a cuboid with
shape 5 9 7 by inserting the times function between each
pair of numbers. With reduce-first, this can be simplified:

5×9×7
315

×⌿5 9 7
315

The function times, together with the operator reduce-first,
creates the derived function ×⌿, recognised as the function
product. Similarly, the derived function +⌿ is the function
sum:

1+2+3+4
10

+⌿1 2 3 4

10

This highlights the versatility of APL in that operators com-
bine with a variety of functions. Another source of versatil-
ity in APL comes from how functions get applied to arrays
of different ranks.

The operator reduce-first ⌿ gets its name by contrast with
the operator reduce /, given that the two operators differ in
the axis along which their derived functions operate. With
the help of the function index generator ⍳ and the left arrow
← that performs assignment, we can create a matrix mat
with shape2 3 and demonstrate the difference between the
two derived functions +/ and +⌿:

mat ← 2 3⍴⍳6
mat

0 1 2
3 4 5

+/mat
3 12

+⌿mat
3 5 7

Plus-reduce-first +⌿ sums along the first axis of its argument
and plus-reduce +/ sums along the last axis of its argument.
For higher-rank arrays, an arbitrary axis can be specified
with the axis operator [axis]. For example, the construct
+/[0] uses reduce with axis to replicate the behaviour of
+⌿.

On top of monadic operators, that take a single operand
on the left, APL provides a series of dyadic operators that
take a left operand and a right operand. One such dyadic
operator is the inner product . (dot), which we use thor-
oughly for the derived function matrix product +.×10. We
exemplify matrix product below:

X ← 2 3⍴0 0 0 1 10 100
Y ← 3 2⍴1 2 3 4 5 6
X Y

┌────────┬───┐
│0 0 0│1 2│
│1 10 100│3 4│
│ │5 6│
└────────┴───┘

X +.× Y
0 0

531 642
Y +.× X

2 20 200
4 40 400
6 60 600

APL functions can only take arrays as arguments, but APL
operators can take functions or arrays as operands. The op-
erator rank ⍤ is one such operator, which takes the forms X

10Presenting the operator inner product in all its generality is outside the
scope of this paper.

(f⍤A) Y and (f⍤A) Y, where X and Y are arbitrary ar-
rays, A is a scalar or a one-item vector (or a two-item vector
if X is present), and f is a function. The derived function is
such that, instead of operating on the full argument(s), op-
erates on subarrays of the specified rank(s) specified in A:

mat ← 2 4⍴⍳8
mat

0 1 2 3
4 5 6 7

mat (×⍤1 0) 1 ¯1
0 1 2 3

¯4 ¯5 ¯6 ¯7
Thematrix mat has two subarrays of rank one, its rows; and
the vector 1 ¯1 has two subarrays of rank zero, its items,
thus(×⍤1 0)will multiply the rows of matwith the items
of 1 ¯1, resulting in a matrix that has the same first row
and a negated second row as mat.

2.3.4 User-defined Functions and Operators. In APL,
one can use the direct functions (dfns) syntax to create user-
defined functions, which can then be named and reused through-
out the APL programs. A dfn is enclosed by braces {} and it
can only take a right argument, or a left and right argument.
Inside a dfn, we use omega ⍵ to refer to the right argument
and alpha ⍺ to refer to the left argument. We provide a short
example:

vec ← 0 1 2 3
⍝ Sum of vec divided by its tally
(+⌿vec)÷≢vec

1.5
⍝ Sum of arg divided by its tally
avg ← {(+⌿⍵)÷≢⍵}
avg vec

1.5
Similarly, the direct operators (dops) syntax can be used to
create user-defined operators. A dop is also enclosed by
braces {} and it can only take a left operand ⍺⍺ if it is a
monadic operator, or a left ⍺⍺ and a right ⍵⍵ operand if it
is a dyadic operator. Inside a dop, ⍵ and ⍺ still refer to the
arguments of the derived function. For example, given a
dyadic function f and a monadic function g, the pattern (g
X) f g Y11 can be abstracted away with the dop {(⍵⍵
⍺) ⍺⍺ ⍵⍵ ⍵}12.

3 Implementation
3.1 Overview
Our implementation of u-net can be roughly divided into
two significant considerations: the implementation of the
fundamental vocabulary of neural networks, and the wiring
of those operations into the actual u-net architecture. We
11A helpful interpretation of this pattern is “preprocess the arguments to
f with the function g”.
12This is a partial model of the operator over ⍥ from APL.

leveraged significant features of APL to implement both as-
pects of the system, and so we will treat each in their own
sub-section.

Additionally, because Co-dfns does not yet support the
complete array of Dyalog primitives and their semantics,
some of the implementation techniques that we use could
be significantly enhanced through the use of a more rich
feature-set. The effect of using these richer features is an
increase in concision and clarity, but we expect that such
improvements would not significantly affect the overall per-
formance of the code, either positively or negatively. We
believe that the overall structure of the code is clear and
simple enough at the current state of Co-dfns to warrant
inclusion almost verbatim here, rather than use the richer
features and require the reader to translate those into the
Co-dfns supported feature set in order to execute them.

One area that deserves particular attention is the design
of APL as a language itself and the specific features that im-
mediately present themselves as particularly well-suited to
expressing neural network computations. Our exploration
of these features uncovered a particular design tension that
is worth discussing in detail. A complete copy of the code
discussed in this paper is included in the appendix.

3.2 Design of APL Primitives for Neural Networks
Themajority of APL primitives find fundamental use in com-
puting neural networks, which isn’t surprising given the
array-oriented and numerical nature of the domain. How-
ever, the stencil operator, introduced in Dyalog APL [Hui
2017], stands out as the most obviously aligned with con-
volutional neural networks. The J programming language
introduced an alternative implementation of the stencil op-
erator earlier [JSoftware 2014], from which Dyalog derived
inspiration for the implementation of their own stencil op-
erator.

The stencil operator takes a function left operand and a
specification array as a right operand. Given a function 𝑓 as
the left operand and a specification 𝑠 as the right operand,
the stencil operator, written 𝑓 ⌺𝑠 , evaluates to a function that
applies 𝑓 on each sliding window specified by 𝑠 . The two
most common sliding window sizes for stencil in u-net are
3 3 for the convolutions, corresponding to a window size of
3 × 3 and a step of 1 for each dimension, and 2 2𝜌2 for the
max pooling layers and up convolutions, corresponding to
a 2 × 2 window size and a step of 2 for each dimension.

When first implementing a convolution, almost everyone
familiar with Dyalog APL and the stencil operator immedi-
ately comes to some variation of the following expression
for convolving a matrix𝑀 with a kernel 𝐾 :

{+/, 𝐾 × 𝜔}⌺3 3 ⊢ 𝑀 (1)

Recall that 𝐾 × 𝜔 is the pointwise multiplication of kernel
𝐾 with one of the 3 × 3 sliding windows. We write +/, 𝐴

to indicate the sum of all elements of array 𝐴. Thus, the
above stencil computes the 2-D convolution over a matrix
with a given 2-D kernel. Because APL’s stencil operator is
leading axis biased, if we were to instead provide a kernel
𝐾 with shape 3 3𝐶 where 𝐶 is the number of channels in
an array𝐴 of shape𝑀 𝑁 𝐶 , the above expression would still
function appropriately. However, if we wish to continue
to extend this to multiple kernels, that is, multiple output
channels, it is less straightforward to compute. The follow-
ing expression computes the convolution of an array𝐴with
shape𝑀 𝑁 𝐼 using kernels 𝐾 with the shape𝑂 3 3 𝐼 where 𝐼
is the number of input channels and𝑂 the number of output
channels:

𝐾{𝑘 ← 𝛼 ⋄ {+/, 𝑘 × 𝜔}⌺3 3 ⊢ 𝜔}⍤3 ⊢ 𝐴 (2)

The result of the above expression is an array of shape𝑂 𝑀 𝑁 .
We make use here of the rank operator, seen in 2.3.3. The
expression 𝐾 𝑓 ⍤3 ⊢ 𝐴 will divide 𝐾 and 𝐴 into subarrays
each of rank 3, that is, each with 3 dimensions, and apply
𝑓 to the corresponding subarrays of 𝐾 and 𝐴. Thus, in our
expression above, our convolution will be applied over the
entire 𝐴 for each output channel described by the first axis
of 𝐾 , thus applying the original 2-D convolution over arbi-
trary numbers of input channels and output channels.

Unfortunately, the above expression has a number of de-
sign flaws. Firstly, the output has the channel count as the
leading axis, while the expected input is to have the channel
count as the trailing axis. This requires that we perform a
transposition of these dimensions after computing the con-
volution in order to return our result to the input format.
Furthermore, the nested structure of the computation re-
sults in two primary functions of non-primitive complexity,
meaning that a more sophisticated analysis of this function
would be required by a compile-time or run-time implemen-
tation in order to recognize this code.

On principle, APL is at its best when it can concisely de-
scribe operations over large arrays at a time, or large sub-
arrays at a time. In particular, concise APL expressions
are possible when the solution can be expressed as a com-
position of basic APL primitives. However, in the case of
the stencil operator, almost all interesting use cases of the
function come from complex, non-simple, non-primitive left
operands. This is further exacerbated by the need to nest the
stencil operation in an outer rank as above. Additionally,
the input sizes provided to the left operand of the stencil
operator are remarkably small, all things considered. This
guarantees that a naive implementation of stencil will be
inefficient and slow, especially on interpreters.

Dyalog APL can mitigate some of these issues through
the use of idiom recognition. However, we argue that idiom
recognition scales particularly poorly to this case. Idiom
recognition has been implemented for the stencil operator,
and there are a selected number of left operand inputs that

are treated specially, so that their performance can be en-
hanced behind the scenes [Hui 2020]. However, because of
the complex nature of the left operand inputs, recognizing
the useful idioms for the stencil operator is a particularly dif-
ficult task, and does not scale well to these sorts of problems.
For instance, while the above stencil operator is considered
an idiom, the following is not:

{⌈/−⌈/−𝜔}⌺(2 2𝜌2) ⊢ 𝐴 (3)
This expression is one way to implement part of the max
pooling layers in u-net. However, there are many other
obvious variations of this expression that would also have
to be considered for idiom recognition, which would other-
wise be missed even if this particular expression were han-
dled. In the case of the design of the stencil operator, trying
to improve its performance via idiom recognition or even
compiler optimization is a relatively significant task. It is
combinatorial and not compositional.

The result is that significant amounts of code would have
to be implemented and maintained in order to support per-
formance enhancements on the stencil operator, with none
of that work benefiting other parts of an APL runtime.

We instead propose an alternative thatwas first suggested
to us by the late Roger Hui, the stencil function [Hui 2020].
The stencil function is a function whose left argument is the
same as the right operand of the stencil operator, and which
receives the same right argument as the right argument to
the function returned by the stencil operator. A reasonable
definition of the stencil function might be:

𝑆𝐹 ← {{𝜔}⌺𝛼 ⊢ 𝜔} (4)
We found that using the stencil function was in fact, not

only easier to work with and more compositional than the
stencil operator, but that it was also universally faster. That
is, even with the idiom recognition that Dyalog has put into
their interpreter to handle the special cases of the stencil
operator, of which 𝑆𝐹 is one [Hui 2020], using the stencil
function instead of the stencil operator was always at least
as fast or faster, despite idiom recognition for the more com-
plex uses of the stencil operator.

The compositionality of 𝑆𝐹 has performance ramifications,
since it is fundamentally an indexing operation, rather than
a computational operation. This categorical shiftmeans that
it can now be approached using the same sorts of lazy index-
ing and fusion operations that are common for other index-
ing operations, such as transposition. This means that the
use of the stencil function can help to broadly reduce inter-
mediate array generation, and performance enhancements
to indexing will compose well with 𝑆𝐹 . Functions and oper-
ators that are already designed to fuse with lazily indexing
functions can then readily take advantage of such features
to work with the output of 𝑆𝐹 as well, granting performance
enhancements across a wider range of applications without

ever implementing any idiom recognition, which reduces
the amount of specialized code that needs to exist as well as
the programmer burden to maintain such code.

To explore this further, a naive implementation of the
stencil function that did not pad its results was implemented
in Co-dfns and used in the following implementations. In
the following sections, we use ⌺ to mean the stencil function
and not the stencil operator as it appears in Dyalog APL. See
equation (6) for a stencil function implementation of equa-
tion (2) and equation (12) for the corresponding implemen-
tation of equation (3).

3.3 Neural Network Vocabulary
The original u-net paper uses five distinct operations to de-
scribe the network (see figure 1):

1. A 3×3 convolution with a ReLU activation function is
used as the primary operation

2. A copy and crop operation is used to transfer data
across one row of the network

3. Max pooling layers on a 2×2 window are used to com-
pute “down” the network

4. A 2×2 transposed convolution goes back “up” the net-
work

5. The final output has a single 1×1 convolution with a
soft-max layer

In our implementation, we mirror this vocabulary by imple-
menting the forward and back functions for each of these
layers, one for each of the above operations. This results in
a total of 10 functions grouped into 5 pairs, which we will
take in turn.

3.3.1 Convolution (3×3) with ReLU. The primary u-net
convolutional layer is a 3×3 convolution with a ReLU acti-
vation function. The convolution in the paper uses “valid”
convolutions, meaning that no padding is used. This implies
that the convolution dimensions of the output array shrink
by 2 for each dimension compared to the input. We define
the forward propagation function 𝐶𝑉 as a function over a
set of kernels ⍺ and a layer ⍵ that obeys the following shape
invariant:

𝜌 𝛼 𝐶𝑉 𝜔 ↔ (−2 + 2 ↑ 𝜌𝜔), −1 ↑ 𝜌𝛼 (5)

We write 𝑆 ↑ 𝐴 to describe the array derived from 𝐴 whose
shape is equal to the shape of 𝐴 except that the leading di-
mensions of 𝑆 ↑ 𝐴 are |𝑆 (absolute value over 𝑆), read as
“the S take of A.” Negative values in 𝑆 take from the “far” or
“trailing” side of the dimension. Thus, the resulting shape of
𝛼 𝐶𝑉 𝜔 is the leading dimensions of the input 𝜔 subtracted
by 2 catenated with the final dimension (the output chan-
nels) of kernel 𝛼 . In the case of a u-net layer, we have in-
put kernels of shape 3 3 𝐼 𝑂 and input layer of shape 𝑁 𝑀 𝐼
where 𝑁 𝑀 are the image/layer dimensions, and 𝐼 𝑂 are the

input and output channel counts, respectively. The result-
ing output layer has shape (𝑁 − 2) (𝑀 − 2) 𝑂 .

Using the stencil function, we define 𝐶𝑉 as follows for
rank 4 kernel inputs and rank 3 layer inputs:

𝐶𝑉 ← {0⌈(,⍤3 ⊢ 3 3⌺𝜔)+.×, [𝜄3]𝛼} (6)
We include the ReLU function 0⌈𝜔 as the final operation fol-
lowing the convolution. We write ,⍤3 ⊢ 𝜔 to describe the
value 𝜔 with its 3 trailing dimensions collapsed into a sin-
gle dimension. We write , [𝜄3]𝜔 to describe the value𝜔 with
its leading 3 dimensions likewise collapsed. In 2.3.3 we pre-
sented +.× as the matrix product form of the operator inner
product, and in 𝐶𝑉 we use its extension to arrays of arbi-
trary rank.

Inside of the u-net architecture itself, we want to save
the output of the convolution and the input to facilitate the
backpropagation pass, and we obtain our kernels from a sin-
gle source containing all network kernels. This results in the
following source code implementation of 𝐶𝑉 that does the
appropriate saving of layers and extracting of kernel data
given a label index into the network as its left argument in-
stead of a kernel, where 𝑍 is our storage for back propaga-
tion and𝑊 contains the weights:
CV←{
z←(,⍤3⊢3 3⌺⊃Z[⍺]←⊂⍵)+.×,[⍳3]⍺⊃W
0⌈z⊣Z[⍺]←⊂Z[⍺],⊂z

}
Computing the backpropagation uses very similar approaches.
Given the output layer 𝑧, input layer 𝑥 , activation layer 𝑎,
weights 𝛼 , and the gradient backpropagated so far 𝜔 , we
compute the transposed weights 𝑤 , the derivative output
layer Δ𝑧 , the weight gradient Δ𝑤 , padded output layer Δ𝑍 ,
and the resulting back gradient Δ𝑥 as follows:

𝑤 ← ⊖ : [1]0 1 3 2 ; 𝛼

Δ𝑧 ← 𝜔 × 0 < 𝑎 (7)
Δ𝑤 ← 3 0 1 2 ; (;, [𝜄2]Δ𝑧)+.×, [𝜄2]3 3⌺𝑥 (8)
Δ𝑍 ← −2 ⊖ −2 : [1] (4 + 2 ↑ 𝜌Δ𝑧) ↑ Δ𝑧
Δ𝑥 ← (,⍤3 ⊢ 3 3⌺Δ𝑍 )+.×, [𝜄3]𝑤 (9)

Since our stencil function does not pad its results, the ex-
pression −2⊖ −2: [1] (4+ 2 ↑ 𝜌Δ𝑧) ↑ Δ𝑧 expands the shape
of Δ𝑧 to ensure that the output convolution dimensions are
2 greater than those of Δ𝑧, where the functions ⊖ and :
are functions to rotate an array. The function :[1] rotates
an array along the 1𝑠𝑡 dimension while ⊖ rotates along the
leading axis. The resulting function Δ𝐶𝑉 is written as fol-
lows:
∆CV←{
x←⊃⍺⊃Z ⋄ a←1⊃⍺⊃Z ⋄ k←⍺⊃W
w←,[⍳3]⊖⌽[1]0 1 3 2⍉k
∆z←⍵×0<a

9

∆Z←¯2⊖¯2⌽[1](4+2↑⍴∆z)↑∆z
∆w←⍺ ∆ 3 0 1 2⍉(⍉,[⍳2]∆z)+.×,[⍳2]3 3⌺x
∆x←w+.×⍨,[2+⍳3]3 3 SF ∆Z

}

In the above code, we have a function Δ that updates the
weights in𝑊 , described in section 3.4.

3.3.2 Copy and Crop. Conceptually, the copy and crop
operation is the simplest of the functions in u-net. Its sole
job is to take the output from one side of the U-shaped net
and move it over to the other side, adjusting the dimensions
to ensure that it fits. In the forward direction, the input layer
will have a greater dimension than the output layer, so we
crop as evenly as possible around the edges and then cate-
nate the result at the head of the layer coming “up” from the
network to form the output layer with twice the channels
of the “up” layer. The following function 𝐶𝐶 computes the
crop of 𝛼 catenated with𝜔 using ↓ (read as “drop”), which is
the opposite function of the previously described ↑ (“take”)
function:

𝐶𝐶 ←
{
𝑝 ←((𝜌𝛼) − 𝜌𝜔) ÷ 2), 𝜔
((⌊𝑝) ↓ (−⌈𝑝) ↓ 𝛼

}
(10)

For dimensions that are not evenly divisible by two, we choose
to round up on the right and bottom sides and round down
on the left and upper sides of the layer. Computing the back-
propagation of 𝐶𝐶 given the input 𝛼 and output gradient 𝜔
simply reverses this operation and expands the shape back
to the original input size. This result is then added to the
appropriate layer in the u-net architecture described in sec-
tion 3.4.

𝑛𝑚 ← −⌊(2 ↑ (𝜌𝛼) − 𝜌𝜔) ÷ 2
Δ𝑥 ← 𝑛 ⊖𝑚 : [1] (𝜌𝛼) ↑ 𝜔 (11)

This leads to the following code for the forward and back-
propagation passes:
CC←{

⍵,⍨(⌊p)↓(-⌈p)↓(⍺⊃Z)⊣p←2÷⍨(⍴⍺⊃Z)-⍴⍵
}

∆CC←{
x←⍺⊃Z ⋄ ∆z←⍵ ⋄ d←-⌊2÷⍨2↑(⍴x)-⍴∆z
(⊃d)⊖(1⊃d)⌽[1](⍴x)↑∆z

}

3.3.3 Max Pooling. Max pooling is a shrinking convolu-
tion that computes themaximumvalue in a non-overlapping
sliding window. Given the stencil function, the max pool
over a layer is given by the following expression:

⌈/−[2], [2 3] (2 2𝜌2)⌺𝜔 (12)

Here we write ⌈/−[2], [2 3]𝜔 to describe an array where we
have collapsed dimensions 2 and 3 and computed the maxi-
mum value reduction over the resulting dimension. For ex-
ample, given an input layer 𝜔 of shape 𝑁 𝑀𝐶 , the result of
(2 2𝜌2)⌺𝜔 is a rank 5 array of shape (𝑁 ÷ 2) (𝑀 ÷ 2)2 2𝐶 .
We then collapse the 2𝑛𝑑 and 3𝑟𝑑 dimensions to form an ar-
ray of shape (𝑁 ÷ 2)(𝑀 ÷ 2)4𝐶 and subsequently find the
maximum value for each vector along the 2𝑛𝑑 dimension,
resulting in an array of shape (𝑁 ÷ 2)(𝑀 ÷ 2)𝐶 .

Computing the backpropagation of this involves replicat-
ing each of the stencil dimensions, which are the two lead-
ing axes in our implementation. We write 𝑛/−𝐴 and 𝑛/[1]𝐴
to indicate the array 𝐴 with each element duplicated or re-
peated along the first and second axes, respectively, 𝑛 times.
Given an input 𝛼 and output layer 𝜔 the following expres-
sion computes the backpropagation:

𝑦 × 𝛼 = 𝑦 ← (𝜌𝛼) ↑ 2/−2/[1]𝜔 (13)
This leads to the following linked source implementation for
max pooling:
MX←{
⌈⌿[2],[2 3](2 2⍴2)⌺⊃Z[⍺]←⊂⍵

}

∆MX←{
x←⍺⊃Z ⋄ ∆z←⍵
y×x=y←(⍴x)↑2⌿2/[1]∆z

}

3.3.4 Transposed Convolution (2×2). In the initial ex-
ploration of this implementation, the upsampling compu-
tation with convolution proved to be the most subtle and
challenging, mostly in part to the opaqueness of implemen-
tations. The u-net paper was not immediately transparent
regarding the exact operations used for this layer and there
were a number of potential design decisions that could have
been made. Moreover, for users reading about upsampling
through convolutions, the descriptions are also the furthest
removed from a reasonable implementation of the same. How-
ever, once the intuition of how an upsampling convolution
matches the shape and form of a non-overlapping sliding
window in the output layer, expressed via the simple ex-
pression 𝐾 ⊂ ⍤ × ⍤2 0 ⊢ 𝐴, the computation becomes much
clearer.13

For this convolution, we change the anticipated kernel
shape from that used for 𝐶𝑉 above. Whereas 𝐶𝑉 expects
kernels of shape 3 3 𝐼 𝑂 , our transposed convolutions expect
kernels of shape 𝐼 2 2𝑂 for input channels 𝐼 and output chan-
nels𝑂 . Given a layer of our standard shape𝑁 𝑀 𝐼 , this gives
the following definition for the upsampling pass.:

𝑈𝑃 ← {,[𝜄2] ,[2 3]𝜌0 2 1 3 4 ; 𝜔 +.×𝛼} (14)
13https://mathspp.com/blog/til/033

The key change here from the reliance on +.× with 𝐶𝑉 is
the use of a dyadic transpose 0 2 1 3 4;. Dyadic transpose is
sometimes considered a somewhat challenging concept in
APL. In brief, for a rank 𝑟 array 𝐴 of shape 𝑆 and targeting
array 𝐷 where (𝜌𝐷) ≡ 𝜌𝑆 , we write 𝐷 ; 𝐴 to describe a
transposed array with shape 𝑇 where 𝑇 [𝐷] ← 𝑆 , assuming
that ∧/−𝐷 ∈ 𝜄𝜌𝑆 , that is, all dimensions of 𝑆 are mentioned
in 𝐷 . So, given a targeting array 0 2 1 3 4 and an input array
𝐴 of shape 𝑁 𝑀 2 2𝑂 , the expression 0 2 1 3 4 ;𝐴 describes
an array with elements from 𝐴 of shape 𝑁 2𝑀 2𝑂 . As the
final operation, we collapse the first two pairs of leading
dimensions, giving a final output array of shape (𝑁×2) (𝑀×
2)𝑂 .

To compute the backpropagation pass, we compute the
convolutions on a 2 × 2 sliding window with stride 2.

Δ𝑤 ← (; ,[𝜄2]𝑥) +.× ,[𝜄2] (2 2𝜌2)⌺Δ𝑧 (15)
Δ𝑥 ← (,[2 + 𝜄3] (2 2𝜌 2)⌺Δ𝑧) +.×; ⍪𝛼 (16)

This gives the following source implementations for trans-
posed convolutions:
UP←{

Z[⍺]←⊂⍵
,[⍳2],[2 3]⍴0 2 1 3 4⍉⍵+.×⍺⊃W

}

∆UP←{
w←⍺⊃W ⋄ x←⍺⊃Z ⋄ ∆z←⍵ ⋄ cz←(2 2⍴2)⌺∆z
∆w←⍺ ∆(⍉,[⍳2]x)+.×,[⍳2]cz
∆x←(,[2+⍳3]cz)+.×⍉⍪w

}

3.3.5 Final 1×1 Convolution. The final convolution is a
1×1 convolution with 2 output channels, which means that
it collapses the final incoming channels into an output layer
with only two channels. This gives the trivial simplification
of our convolution code over layer 𝜔 and kernel 𝛼 :

𝜔 + . × 𝛼 (17)
Additionally, the paper describes using a soft-max layer, which
we include at this phase:

1𝑒−8 + 𝑧 ÷ [𝜄2]+/𝑧 ← ∗𝜔 − [𝜄2] ⌈/𝜔 (18)
Computing the backpropagation is likewise a simplifica-

tion of the more complex 𝐶𝑉 code:

Δ𝑤 ← (; ,[𝜄2]𝑥)+.× ,[𝜄2]Δ𝑧 (19)
Δ𝑥 ← Δ𝑧 +.×;𝑤 (20)

Which leads to the following source implementations:
C1←{

Z[⍺]←⊂⍵
1E¯8+z÷[⍳2]+/z←*z-[⍳2]⌈/z←⍵+.×⍺⊃W

}

∆C1←{
w←⍺⊃W ⋄ x←⍺⊃Z ⋄ ∆z←⍵

∆w←⍺ ∆(⍉,[⍳2]x)+.×,[⍳2]∆z
∆x←∆z+.×⍉w

}

3.4 U-net Architecture
Given the core vocabularies defined in section 3.3, the re-
maining challenge with implementing u-net is to link to-
gether the appropriate layers and compositions to form the
complete network as described by figure 1. To do this, we
observe that the structure of the u-net diagram is an almost
symmetric pattern. The output layer computations form 3
operations which are not part of the pattern, but the rest
of the pattern decomposes into 4 depths, each with 6 op-
erations each. table 2 contains a visual arrangement of the
kernel shapes used in our architecture mirroring the overall
structure of figure 1.

Additionally, we note that the U-shaped structure also
mimicks the down and up nature of a recursive program call-
tree. Thus, our overall strategy is to implement a recursive
function 𝐿𝐴 that receives an index identifying a particular
depth of the network, computes the appropriate “downward
pass” operations before recuring deeper into the network
and finally computing the upwards passes on the return of
its recursive call. We likewise implement backpropagation
in same way, but in the opposite direction. Assuming that
𝛼 contains the computed depth offset for the network layer,
we write 𝛼 + 𝑖 to access the 𝑖th column of the network de-
scribed in table 2 at the depth 𝛼 ÷ 6.

Our forward pass function is responsible for initializing
an appropriate holding place for the intermediate results
produced by forward propagation for use by the backprop-
agation function. Additionally, after the recursive compu-
tation, there are the final three operations, 𝐶1 and two 𝐶𝑉
operations, that must be called before returning. We also
assume that we may receive a rank 2 matrix instead of a
rank 3 layer as input, and so we reshape the input to ensure
that we always have a rank 3 input to 𝐿𝐴. This gives us the
following function definition:

FWD←{
Z⊢←(≢W)⍴⊂⍬
⍝ Forward propagation layers ...
LA←{

⍺≥≢Z:⍵
down←(⍺+6)∇(⍺+2)MX(⍺+1)CV(⍺+0)CV ⍵
(⍺+2)CC(⍺+5)UP(⍺+4)CV(⍺+3)CV down

}
2 C1 1 CV 0 CV 3 LA ⍵⍴⍨3↑1,⍨⍴⍵

}
11

OpeRation
𝐶𝑉 𝐶𝑉 𝑀𝑋 𝐶𝑉 𝐶𝑉 𝑈𝑃

D
ep

th

0 3 3 1 64 3 3 64 64 0 0 64 64 3 3 256 128 3 3 128 128 128 2 2 64
1 3 3 64 128 3 3 128 128 0 0 128 128 3 3 512 256 3 3 256 256 256 2 2 128
2 3 3 128 256 3 3 256 256 0 0 256 256 3 3 1024 512 3 3 512 512 512 2 2 256
3 3 3 256 512 3 3 512 512 0 0 512 512 3 3 512 1024 3 3 1024 1024 1024 2 2 512

Downward Pass Upward Pass

Table 2. A rectangular arrangement of the u-net network

The backwards computationmirrors this pattern, except that
it proceeds in the opposite direction and also defines an up-
dater function Δ that will update the network weights in
𝑊 and the velocities in 𝑉 based on a given gradient 𝜔 and
index 𝛼 pointing to a specific location in the network.
BCK←{

Y←⍺ ⋄ Y∆←⍵
∆←{

V[⍺]←⊂⍵+MO×(⍴⍵)⍴⍺⊃V
W[⍺]←⊂(⍺⊃W)-LR×⍺⊃V

}
⍝ Back propagation layers ...
∆LA←{

⍺≥≢Z:⍵
d←{(⍺+3)∆CV(⍺+4)∆CV(⍺+5)∆UP ⍵}
c←{(⍺+2)∆CC ⍵}
u←{(⍺+0)∆CV(⍺+1)∆CV(⍺c⍵)+(⍺+2)∆MX}
⍺ u (⍺+6)∇ ⍺ d ⍵↑[2]⍨-2÷⍨⊃⌽⍴⍵

}
3 ∆LA 0 ∆CV 1 ∆CV 2 ∆C1 Y∆-(~Y),[1.5]Y}

We also need to compute an error over the soft-max com-
puted by 𝐹𝑊𝐷 . This is given by the following function,
which is based off of the error function given in the origi-
nal u-net paper by Ronneberger et al. [2015].
E←{-+⌿,⍟(⍺×⍵[;;1])+(~⍺)×⍵[;;0]}
Finally, we wire all of these functions together into a 𝑅𝑈𝑁
function that runs the forward pass and backward pass func-
tions and returns three values, the expected inputs 𝑌 , the
computed results 𝑌Δ from 𝐹𝑊𝐷 , and the error given by
𝑌 𝐸 𝑌Δ. We reshape the original reference input to match
the size of 𝑌Δ.
RUN←{

Y∆←FWD ⍺
Y←⌊0.5+nm↑⍵↓⍨2÷⍨(⍴⍵)-nm←2↑⍴Y∆
Y Y∆(Y E Y∆)⊣Y BCK Y∆

}

4 Performance
To examine the performance profile of our APL implemen-
tation, we primarily focused on comparing our u-net im-
plementation against a reference implemented in PyTorch

[Paszke et al. 2019], which is an easy to use Python frame-
work with good performance. In addition to this primary
performance analysis, we examined the performance of two
varieties of stencil computations within the APL language.
We also make note of some small exploratory effects that
we discovered while implementing the stencil function and
convolutions in Co-dfns.

4.1 U-net Performance
We were primarily interested in the costs of executing a sin-
gle run of the u-net over a single image source in both the
forwards and backwards directions. We compared perfor-
mance over the following platforms:
• Dyalog APL 18.0 64-bit Windows interpreter
• Co-dfns (v4.1.0 master branch) using the CUDA back-

end (AF v3.8.1)
• Co-dfns (v4.1.0 master branch) using the CPU back-

end (AF v3.8.1)
• PyTorch v1.10.2 with the CUDA gpu backend
• PyTorch v1.10.2 with the multi-threaded CPU back-

end
• PyTorch v1.10.2 with the single-threaded CPU back-

end
The results of the execution can be seen in figure 2. The
timings do not include the cost of reading the image data
from disk, but they do include the costs of transferring the
image input data and the resulting error and forward propa-
gation results back to the CPUmain memory. In our testing,
data transfer costs in Co-dfns accounted for significantly
less than 5% of the total runtime.

The hardware used was an NVIDA GeForce RTX 3070
Laptop GPU with 8GB of dedicated graphics memory. We
used NVIDIA driver version 511.65. The CPU was an In-
tel Core i7-10870H with 16 logical cores @ 2.2GHz. Main
system memory was 32GB of DDR4 RAM. The system was
running an up to date release of Microsoft Windows 11.

As input we used the original image data from the ISBI
benchmark referenced in the u-net paper [Cardona et al.
2010; Ronneberger et al. 2015]. These images are 512 × 512
images in grayscale with a binary mask for training. Each
run took one of these images and associated training mask

12

Figure 2. Performance results for U-net across a range of platforms

0 1 2 3 4 5 6 7 8 9 Avg

Dyalog 80.90 77.56 78.32 81.33 77.95 80.51 81.08 81.87 80.23 79.42 79.92
Co-dfns (CPU) 34.11 33.66 35.98 37.72 38.20 38.64 38.11 38.46 38.57 38.76 37.22
Co-dfns (GPU) 8.39 8.20 8.62 8.63 8.49 8.39 8.19 8.63 8.51 8.59 8.46
PyToRch (CPU) 17.27 17.34 17.21 16.60 17.46 17.01 17.33 17.04 17.36 17.33 17.22
PyToRch (SMP) 6.31 6.42 6.45 6.19 6.67 6.49 6.43 6.38 6.36 6.55 6.42
PyToRch (GPU) 3.50 3.50 3.47 3.44 3.44 3.44 3.44 3.44 3.46 3.46 3.46

Table 3. Raw timings for each computing platform in seconds

and computed the result of forward and backwards propa-
gation and the error as well as updating the weights for the
network.

Whenworking on the network, APL implementations gen-
erally do not have a concept of small floating point values.
Rather, their default is to always use 64-bit floating point
values when floats are called for. In order to try to mimic
this behavior as closely as possible, we attempted to feed 64-
bit data into the PyTorch models. However, because of the
opaqueness of the PyTorch implementation, we were not
able to fully verify that 64-bit values are used throughout the
PyTorch computational network. On the other hand, the re-
liance on 64-bit only floating points, while a boon to conve-
nience and user-friendliness for non-computer science pro-
grammers, creates well-defined performance issues for an
application like this.

When running the benchmark, we computed the average
of 10 runs, ensuring that we discarded the first run each

time, since these runs often contained significant setup and
bootstrapping code (PyTorch’s optimizer, the JIT optimiza-
tion in Co-dfns, and so forth). The figure includes informa-
tion about the variance of the individual runs as well as the
average run time in seconds.

Examining the data, it is clear why traditional APL imple-
mentations were relatively unsuited to extensive use within
themachine learning space. Dyalog’s interpreter preformed
the slowest by a very large magnitude. After this, the single-
threaded CPU implementations in Co-dfns and PyTorch are
predictably the next slowest, with the Co-dfns implemen-
tation running about a factor of 2.2 times slower than the
equivalent PyTorch implementation.

When acceleration techniques are employed, the differ-
ences in execution speed begin to shrink, with PyTorch’s
multi-threaded and GPU-based implementations coming in
fasted, and Co-dfns’ GPU backend running at roughly 2.4
times slower than the PyTorch GPU execution.

We observed the widest variance in performance results
in the Co-dfns CPU and Dyalog interpreter-based runs, and
very little variance in the GPU-based runs or in PyTorch
itself.

4.1.1 Co-dfnsRuntime Implementation. Thestencil func-
tion was modeled in APL and used to conduct the above
benchmark. The model, written in APL, is a naive imple-
mentation of the stencil function and contains no special
optimizations other than to distinguish between slidingwin-
dows of step 1 and non-overlapping, adjacentwindows (such
as used for the max pooling layers). Additionally, no spe-
cialized code was used within Co-dfns that was specific or
specialized to neural network programming.

The above benchmark therefore represents a comparison
of the PyTorch implementation against a naive and unspe-
cialized implementation in APL executed with the general-
purpose runtime used in Co-dfns that provides generalized
GPU computation but does not include domain-specific op-
timizations such as those available in PyTorch.

4.2 APL Stencil Primitives
In this section we present some benchmarks relative to the
speedup we get when considering convolutional layers and
max pooling layers that are based on the stencil function
instead of the stencil operator.

Figure 3 compares convolutional layers based on the sten-
cil function with convolutional layers based on the stencil
operator. To produce this benchmark, we consider inputs of
different sizes and number of channels. Then, for each value
𝑠𝑖𝑧𝑒 and number 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 , we create a random input cuboid
of shape 𝑠𝑖𝑧𝑒 𝑠𝑖𝑧𝑒 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 that is passed through a convolu-
tional layerwith a kernel having shape 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 3 3 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 ,
which means the output also has shape 𝑠𝑖𝑧𝑒 𝑠𝑖𝑧𝑒 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 .
After creating these inputs and kernels, we benchmark the
runtime of the two convolutional layers (one stencil operator-
based and one stencil function-based) and divide the run-
time of the convolutional layer based on the stencil func-
tion by the runtime of the convolutional layer based on the
stencil operator, in order to compute the speedup we get by
adopting the stencil function.

Similarly, Figure 4 compares max pooling layers based on
the stencil function with max pooling layers based on the
stencil operator. The experimental setup is identical, except
that we do not have to generate random kernels for the max
pooling layers.

While Figure 3 shows a significant speedup achieved through
the introduction of the stencil function in the convolutional
layers, Figure 4 shows that the max pooling layer based on
the stencil function is only slightly faster when the num-
ber of channels is small, becoming slightly slower than the
convolutional layer based on the stencil operator when the
number of channels increases.

4.3 Microbenchmarks Against Other Libraries
It is worth noting that we also explored optimizing our u-
net with specialized functions available via the Co-dfns plat-
form (specificmax filters and convolutions) that are domain-
specific operations much like those available in PyTorch.
This sort of operation can be done via “idiom recognition”
and for our small convolution expressions, it is quite con-
ceivable that idiom recognition could apply and convert these
functions to use domain-specific code under the hood for
convolutions and maximum filters, &c.

However, we aborted this line of inquiry for now because
inmicrobenchmarking the domain-specific functions against
our naive implementation of the stencil function, we discov-
ered that the domain specific functions were actually a fac-
tor of 2 slower than our naive implementations in our pri-
mary sample inputs. Given that the underlying specialized
functions are supposed to wrap the CuDNN library [Yurke-
vitch 2020], it is surprising that we achieved faster results
with the naive stencil function implementation over the do-
main specific implementations available within the general
purpose array libraries leveraged by Co-dfns.

We are not sure of the cause of these slowdowns, and
therefore, we did not include a formal benchmark of these
results here. It is possible that a misconfiguration or some
other element is causing these degradations, and so we in-
tend to explore these performance issues in more detail, but
we wished to at least note this effect, since this represents
a different result from our benchmarking here, which tests
Co-dfns against a specialized framework, rather than amore
generalized array frameworkwith specialized functionswithin
it. The specialized frameworks appear to be clearly faster at
the moment than the naive Co-dfns implementations, but
this is not true thus far in our limited testing of specialized
functions exposed within more generalized array libraries.

5 Discussion
5.1 Pedagogy
Pedagogy is a concern for new approaches to solving prob-
lems both in academic as well as industrial spaces. We be-
lieve that APL is a double-edged sword in this regard. On
one hand, there is significant institutionalmomentum around
languages like Python. This creates a large base of prior
knowledge which can be leveraged by new users. This re-
sults in users feeling like learning a Python based frame-
work is easier than learning awhole new language, and they
are probably right, in the short term.

However, it has been argued [Iverson 2007] that APL has
two distinct advantages from a pedagogical point of view
that may warrant more interest. First, what one learns in
APL tends to also have direct skills transferrence to many
other programming domains, whereas in a more domain-
specific, library-centric approach, learning the particular API
for one domain often does not transfer any skills beyond

Figure 3. Speedup of stencil function-based convolution with respect to stencil operator-based on inputs of shape 𝑠𝑖𝑧𝑒 ×
𝑠𝑖𝑧𝑒 × 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 and kernels of size 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 × 3 × 3 × 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 .

0.98

1.04

0.99

0.96

0.96

0.87

1.1

1.06

0.99

1

0.85

0.8

1.12

1.08

1

0.86

0.89

0.8

1.02

1

0.96

0.93

0.88

0.83

1.2

1.22

1.14

1.05

0.96

0.85

16

32

64

128

256

512

32 64 128 256 512

Feature map size

#
 o

f 
c
h

a
n

n
e

ls

0.8
0.9
1.0
1.1
1.2

speedup

Stencil function vs stencil operator max pooling speedup

Figure 4. Speedup of stencil function-based max pooling with respect to stencil operator-based on inputs of shape 𝑠𝑖𝑧𝑒 ×
𝑠𝑖𝑧𝑒 × 𝑐ℎ𝑎𝑛𝑛𝑒𝑙𝑠 .

that domain directly. In the case of u-net, all of the op-
erations used to build the u-net system are general array
programming concepts that are widely applicable to many
other domains, and are not restricted solely to convolutional
neural networks.

Second, a transparent and direction implementation of u-
net in APL is somewhat uniquely compact and simple, mak-
ing it much easier to not only delve deeper into the CNN
domain, but also to make adjustments and modifications
to taste as one becomes more experienced. Starting with

a transparent implementation enables programs to be en-
hanced or adapted or optimized without requiring the inclu-
sion of abstractions that increase program indirection and
opaqueness. The notational aspects of APL facilitates this
sort of expressive power in a way that other languages do
not, especially from a “human factors” perspective.

However, the current learning materials for APL, partic-
ularly in a space like neural networks, are clearly underde-
veloped and in need of improvement. We believe it is likely

possible to make it as easy to reference a convolutional im-
plementation in APL as an API reference for PyTorch, but
the current ecosystem is not there yet.

5.2 Performance
Clearly, specialized frameworks for deep neural networks
are still the best way to go in order to achieve the abso-
lute maximum in performance at present. However, our re-
sults indicate that the gap between reliance on specialized
frameworks and the freedom to use more general purpose
and transferrable programming languages while still achiev-
ing competitive performance is not nearly as large as might
have been the case even a few years ago.

Given that almost zero special optimization is taking place
for the APL implementation executed under the Co-dfns
runtime, it is impressive that we are able to see results that
come close to a factor of 2 of the specialized frameworks.
Given some of the obvious design issues that would con-
tribute to slower performance, it seems more reasonable
to be able to expect more general purpose languages like
APL to be able to reach performance parity with specialized
frameworks, without the requirement that the user learn a
special API, or import specialized dependencies. In more
complex applications that leverage APL for other domain-
intensive work, this suggests that APL might facilitate scal-
ing such applications to integrate machine learning algo-
rithms more easily and with less programmer effort than
might be required to integrate a separate framework like
PyTorch.

5.3 Stencil Operator
Theresultswe found regarding stencil operator performance
vs. stencil function performance suggest to us that compa-
nies like Dyalog or implementors of Co-dfns should focus
on implementing and improving the performance of a sten-
cil function instead of continuing with the use of a stencil
operator, which suffers from a number of design issues that
makes it incongruouswith the rest of an otherwise elegantly
designed language.

It is likely that scalable performance with the stencil func-
tion will be easier to achieve and easier to maintain over the
long term. Moreover, the stencil function results in more
compositional code that it easier to work with using the rest
of the APL language than the stencil operator.

5.4 APL vs. Frameworks
We have demonstrated that APL itself, without libraries or
additional dependencies, is exceptionally well suited to ex-
pressing neural network computations, at a level of inher-
ent complexity that is arguably equal or possibly even less
than that of the reference PyTorch implementation. At the
very least, it is less code with less underlying background
code and layers. This comes at the admittedly heavy cost
of being completely foreign and alien to most programmers

who are more familiar with languages like Python. This cer-
tainly creates a fundamental and immediate learning cost to
APL over other frameworks, since other frameworks can as-
sume a wider range of pre-knowledge around their chosen
language implementation.

It remains unclear, however, whether, if this pre-knowledge
were taken away, APL would represent a compelling value
proposition for such programming tasks or not. Indeed, it
is exceptionally challenging to divorce the present reality
of prior background knowledge from such a question. Even
fundamental knowledge like what it means to do array pro-
gramming and how to structure problems in an array-style
are rarely if ever taught at universities, whereasmost classes
spend significant amounts of time teaching students how to
utilize the Python-style programming model of PyTorch.

The argument that APL may be used more widely and
broadly than PyTorch on a wider range of problems using
the same skillset may not matter to users who are only in-
terested in deep learning algorithms.

APL presently has a higher barrier to entry, but rewards
the user with full and effortless control over what’s being
done in a way that other systems do not. This may present
itself as a distinct advantage to users who are looking to
expand “off the beaten track” and utilize novel approaches
that do not easily fit within existing frameworks.

We encountered significant difficulties in identifying ex-
actly what the original authors did based on their paper
alone because of many implementation details that were
omitted. On the other hand, APL enables us to express our
entire implementation in a way that makes every implemen-
tation detail clear, improving the ability of others to repro-
duce our work.

Finally, in the implementation of u-net in APL, we gained
insights into the architecture that had a direct and mate-
rial influence on the PyTorch reference implementation that
would not have emerged without first having studied the
APL implementation. Thus, we gained significant insight
simply from doing the APL implementation, even if wewere
to re-implement that code in PyTorch.

6 Related Work
Two particular avenues of research warrant particular men-
tion here. In addition to the Co-dfns compiler, Šinkarovs
et al. [2019] have explored alternative implementations to
CNNs, though not u-net specifically. They focused specif-
ically on APL as a productivity enhancement for CNN de-
velopment, and only benchmarked the APL implementation
on the CPU using the Dyalog APL interpreter. However,
they indicated work in progress on a compiled version us-
ing the APEX compiler with a SaC backend. Their conclu-
sion regarding the performance of APL-based systems may
have been premature given the results we found here, but
they make a case for the usability of APL even with the

performance numbers they achieved. While their code ex-
hibits some material differences to that given here, there are
nonetheless some similarities that demonstrate some level
of convergence around implementing CNNs in APL.

Another approach toGPU-based array programmingwith
an APL focus is the TAIL/Futhark system [Henriksen et al.
2017], which is a compiler chain taking APL to the TAIL
(Typed Array Intermediate Language) and then compiling
TAIL code using the Futhark GPU compiler backend. While
the authors are not aware of any work implementing com-
plex neural networks with this chain, it represents an inter-
esting approach to compilation of APL via typed interme-
diate languages, which have the potential to enhance the
fusion that can be done with an operation like the stencil
function.

Other programming environments that are often catego-
rized as array programming environments, such as Matlab
[MathWorks 1992], Julia [Bezanson et al. 2017], and Python/
Numpy [Harris et al. 2020; Van Rossum and Drake 2009],
are not exceptionally performant on their own for machine
learning, but often wrap libraries to do so. Unfortunately,
many of these languages use a syntax thatmuchmore closely
mirrors that of Python than APL. In our perspective, this re-
duces the value proposition of such languages over using
specialized frameworks, since one does not obtain the par-
ticular clarity and productivity benefits associated with the
APL notation.

7 Future Work
One of the most obvious questions to answer in future work
is the reason for the slower performance of the specialized
convolution functions against our naive implementationwhen
using the same backend in Co-dfns.

There are a number of design elements of the current crop
of APL implementations, including Co-dfns, which hamper
performance for machine learning. Especially, the use of
64-bit floating points without any feature to reduce their
size makes memory usage a concern. Additionally, no opti-
mization on the design of stencil has been done, while opti-
mizations related to lazy indexing, batch processing, and a
number of other features seem readily accessible.

Additionally, we would like to explore the potential of us-
ing such systems to improve machine learning pedagogy by
encouraging students to have access to high-performance,
but also transparent, implementations of foundational ma-
chine learning concepts. There are still some challenges to
recommending this approach at scale for a large number
of educational institutions, but we believe work on under-
standing the pedagogical benefits of APL warrants further
research in addition to exploring APL in the professional
space.

8 Conclusion
Given the notational advantages of APL and the concision
and clarity of expression that one can obtain, we explored
the potential impact of using APL as a language for imple-
menting convolutional neural networks of reasonable com-
plexity. We found that, though the traditional implemen-
tations of APL suffer from performance issues that would
prevent widespread use in either academic, educational, or
industrial contexts, compilers such as Co-dfns are capable of
compiling complete neural network programs (in our case,
the u-net architecture) and producing much more competi-
tive performance results (within a factor of 2.2 - 2.4 times of
our reference PyTorch implementation). This is despite the
naive nature of our implementation and the naive optimiza-
tion support for neural networks on the part of the Co-dfns
compiler.

Furthermore, we found that our effort to implement u-net
in APL resulted in a concise but fully unambiguous imple-
mentation that provided transparency over the entire source,
without any frameworks or library dependencies. Despite
being a complete “by hand” implementation, its complex-
ity of expression is on par with that of PyTorch and other
specialized frameworks, or even better, particularly in cases
wheremore exploration and novel implementation is required,
or when customized integrations may be called for. The in-
sights that we gained from implementing u-net in APL af-
fected our implementation of a reference implementation
in PyTorch directly, suggesting that APL may have signif-
icant pedagogical advantages for teaching neural network
programming and machine learning in general.

Appendix A: Complete APL U-net implementation
:Namespace UNET

W←⍬ ⋄ V←⍬ ⋄ Z←⍬ ⋄ LR←1e¯9 ⋄ MO←0.99

FWD←{Z⊢←(≢W)⍴⊂⍬
CV←{0⌈z⊣Z[⍺]←⊂Z[⍺],⊂z←(,[2+⍳3]3 3⌺⊃Z[⍺]←⊂⍵)+.×,[⍳3]⍺⊃W}
CC←{⍵,⍨(⌊p)↓(-⌈p)↓(⍺⊃Z)⊣p←2÷⍨(⍴⍺⊃Z)-⍴⍵}
MX←{⌈⌿[2],[2 3](2 2⍴2)⌺⊃Z[⍺]←⊂⍵}
UP←{((2×¯1↓⍴⍵),¯1↑⍴⍺⊃W)⍴0 2 1 3 4⍉⍵+.×⍺⊃W⊣Z[⍺]←⊂⍵}
C1←{1E¯8+z÷[⍳2]+/z←*z-[⍳2]⌈/z←⍵+.×⍺⊃W⊣Z[⍺]←⊂⍵}
LA←{⍺≥≢Z:⍵

down←(⍺+6)∇(⍺+2)MX(⍺+1)CV(⍺+0)CV ⍵
(⍺+2)CC(⍺+5)UP(⍺+4)CV(⍺+3)CV down}

2 C1 1 CV 0 CV 3 LA ⍵⍴⍨3↑1,⍨⍴⍵}

BCK←{Y←⍺ ⋄ Y∆←⍵
∆←{0⊣W[⍺]←⊂(⍺⊃W)-LR×⊃V[⍺]←⊂⍵+MO×(⍴⍵)⍴⍺⊃V}
∆CV←{w←,[⍳3]⊖⌽[1]0 1 3 2⍉⍺⊃W ⋄ x←⊃⍺⊃Z ⋄ ∆z←⍵×0<1⊃⍺⊃Z
∆Z←¯2⊖¯2⌽[1](4+2↑⍴∆z)↑∆z
_←⍺ ∆ 3 0 1 2⍉(⍉,[⍳2]∆z)+.×,[⍳2]3 3⌺x
w+.×⍨,[2+⍳3]3 3⌺∆Z}

∆CC←{x←⍺⊃Z ⋄ ∆z←⍵ ⋄ d←-⌊2÷⍨2↑(⍴x)-⍴∆z ⋄ (⊃d)⊖(1⊃d)⌽[1](⍴x)↑∆z}
∆MX←{x←⍺⊃Z ⋄ ∆z←⍵ ⋄ y×x=y←(⍴x)↑2⌿2/[1]∆z}
∆UP←{w←⍺⊃W ⋄ x←⍺⊃Z ⋄ ∆z←⍵ ⋄ cz←(2 2⍴2)⌺∆z
_←⍺ ∆(⍉,[⍳2]x)+.×,[⍳2]cz
(,[2+⍳3]cz)+.×⍉⍪w}

∆C1←{w←⍺⊃W ⋄ x←⍺⊃Z ⋄ ∆z←⍵ ⋄ _←⍺ ∆(⍉,[⍳2]x)+.×,[⍳2]∆z ⋄ ∆z+.×⍉w}
∆LA←{⍺≥≢Z:⍵
down←(⍺+6)∇(⍺+3)∆CV(⍺+4)∆CV(⍺+5)∆UP ⍵↑[2]⍨-2÷⍨⊃⌽⍴⍵
(⍺+0)∆CV(⍺+1)∆CV(⍵ ∆CC⍨⍺+2)+(⍺+2)∆MX down}

3 ∆LA 0 ∆CV 1 ∆CV 2 ∆C1 Y∆-(~Y),[1.5]Y}

E←{-+⌿,⍟(⍺×⍵[;;1])+(~⍺)×⍵[;;0]}

RUN←{Y Y∆(Y E Y∆)⊣(Y←⌊0.5+nm↑⍵↓⍨2÷⍨(⍴⍵)-nm←2↑⍴Y∆)BCK⊢Y∆←FWD ⍺}

:EndNamespace

Appendix B: PyTorch Reference Implementation
import torch
import torch.nn as nn
import torchvision
import torchvision.transforms.functional

class TwoConv(nn.Module):

def __init__(self, in_channels, out_channels):
super().__init__()

self.path = nn.Sequential(
nn.Conv2d(in_channels, out_channels,

kernel_size=(3, 3), bias=False),
nn.ReLU(inplace=True),
nn.Conv2d(out_channels, out_channels,

kernel_size=(3, 3), bias=False),
nn.ReLU(inplace=True),

)

def forward(self, x):
return self.path(x)

class Down(nn.Module):

def __init__(self, in_channels):
super().__init__()

self.path = nn.Sequential(
nn.MaxPool2d(kernel_size=(2, 2), stride=2),
TwoConv(in_channels, 2 * in_channels),

)

def forward(self, x):
return self.path(x)

class Up(nn.Module):

def __init__(self, in_channels):
super().__init__()

self.upsampling = nn.ConvTranspose2d(
in_channels,
in_channels // 2,
kernel_size=(2, 2),
stride=2,
bias=False,

)
self.convolutions =

TwoConv(in_channels, in_channels // 2)

def forward(self, x_to_crop, x_in):

upped = self.upsampling(x_in)
cropped = torchvision.transforms.functional.center_crop(

x_to_crop, upped.shape[-2:]
)
x = torch.cat([cropped, upped], dim=1)
return self.convolutions(x)

class USegment(nn.Module):

def __init__(self, in_channels, bottom_u=None):
super().__init__()

# Default value for the bottom U.
if bottom_u is None:

bottom_u = lambda x: x

self.down = Down(in_channels)
self.bottom_u = bottom_u
self.up = Up(2 * in_channels)

def forward(self, x):
return self.up(x, self.bottom_u(self.down(x)))

class UNet(nn.Module):

def __init__(self):
super().__init__()

self.u = USegment(512)
self.u = USegment(256, self.u)
self.u = USegment(128, self.u)
self.u = USegment(64, self.u)
self.path = nn.Sequential(

TwoConv(1, 64),
self.u,
nn.Conv2d(64, 2, kernel_size=1, bias=False),

)

def forward(self, x):
return self.path(x)

